

Converting SAS/GRAPH to ODS Graphics

Jim Horne

Lowe's Companies, Inc.

Jim.s.horne@lowes.com

Reason

- As of SAS 9.3, SAS has moved ODS Graphics and the Statistical Graphics procedures from SAS/GRAPH[®] to Base SAS[®]. This provides us an opportunity to eliminate SAS/GRAPH by converting SAS/GRAPH procedures to ODS Graphics procedures

Purpose

- What this presentation is
 - A brief introduction to ODS Graphics and the statements you must use to run it
 - A comparison of simple SAS/GRAPH and ODS Graphics output and the statements that produce it
- What this presentation is not
 - An exhaustive introduction to ODS Graphics
 - An introduction to graphing using SAS. Familiarity with SAS/GRAPH is assumed

Agenda

- Introduction
- SAS Statements
 - SAS/GRAPH statements going away
 - Statements still used
 - New statements
- ODS Graphics Procedures and Graph Template Language (GTL)
- Graph Comparisons
- Extras
- Wrap up

SAS/GRAPH Statements Going Away

- LEGEND
- SYMBOL
- AXIS
- GOPTIONS
- All SAS/GRAPH procedures

Statements Still Required

- FILE
- ODS HTML
- Used just as they are today

Common Statements

- TITLE
- FOOTNOTE
- NOTE
- FORMAT
- LABEL
- BY
- WHERE

ODS GRAPHICS Statement (New)

- ODS GRAPHICS ON/OFF ;
- ODS GRAPHICS ON options ;
- Common options
 - IMAGENAME
 - IMAGEMAP
 - HEIGHT
 - WIDTH
 - RESET

ODS GRAPHICS Statement Examples

```
ODS GRAPHICS ON;
```

```
ODS GRAPHICS OFF;
```

```
ODS GRAPHICS ON / IMAGENAME=xxxx;
```

```
ODS GRAPHICS ON /IMAGENAME=xxxx  
  IMAGEMAP;
```

```
ODS GRAPHICS ON / RESET=ALL;
```

Statistical Graphics Procedures

- SGDESIGN
- SGPANEL
- SGPLOT
- SGRENDER
- SGSCATTER

SGPLOT

- BAND
- BUBBLE
- DENSITY
- DOT
- HBAR/VBAR
- HBOX/VBOX
- HIGHLOW
- HISTOGRAM
- HLINE/VLINE
- NEEDLE
- REFLINE
- REG
- SCATTER
- SERIES
- STEP
- VECTOR
- WATERFALL

SGRENDER

- PROC SGRENDER
 - PROC TEMPLATE
 - Graph Template Language

SGRENDER Example

```
PROC TEMPLATE;  
  DEFINE STATGRAPH minimumreq;  
    BEGINGRAPH;  
      LAYOUT OVERLAY;  
        SCATTERPLOT X=WEIGHT Y=HEIGHT;  
      ENDLAYOUT;  
    ENDGRAPH;  
  END;  
RUN;  
  
PROC SGRENDER DATA=SASHELP.CLASS  
  TEMPLATE=minimumreq;  
RUN;
```

Graph Comparisons

Format of Comparison

- Three slides per comparison
 - SAS/GRAPH view
 - Side by side comparison between SAS/GRAPH and ODS Graphics
 - ODS Graphics view


Statements Used to Build Graphs

- Web page definition
ODS LISTING CLOSE;
FILENAME odsout *"/a-file-definition/"* ;
ODS HTML BODY="odsclass.html" (TITLE="ODS Class")
STYLE=htmlblue
PATH=odsout (URL=NONE) ;
- SAS/GRAPH statement
SYMBOL1 INTERPOL=JOIN;
- ODS Graphics statement
ODS GRAPHICS ON / IMAGENAME='odsclass'
NOBORDER NOANTIALIAS
TIPMAX=3000
HEIGHT=725px ;

Graphs we use today

- Line plots
- Bar charts
 - Simple charts
 - Grouped charts
 - Stacked charts
 - Bar charts combined with line plots
- Scatter plots
- Pie charts
- Stacked Plots

Line Plots



Line Plots


SAS/GRAPH

```
PROC GPLOT DATA=cpr70lp ;  
  TITLE "GPLOT Line Plots" ;  
  PLOT usedgps*time=system ;  
RUN;
```


ODS Graphics

```
PROC SGPLOT DATA=cpr70lp ;  
  TITLE "SGPLOT Line Plots" ;  
  SERIES X=time Y=usedgps /  
    GROUP=system ;  
RUN;
```

Line Plots



Simple Bar Charts



Simple Bar Charts


SAS/GRAPH

```
PROC GCHART DATA=meanengs ;  
BY date DESCENDING shift ;  
  TITLE "GCHART Simple Bar  
  Chart" ;  
  TITLE2 "#BYVAL(shift) Shift" ;  
  HBAR system / DISCRETE  
  SUMVAR=usedgps  
  SUM  
  SUMLABEL = 'Engs'  
;  
RUN;
```


ODS Graphics

```
PROC SGPLOT DATA=meanengs  
  CYCLEATTRS ;  
BY date DESCENDING shift ;  
  TITLE "SGPLOT Simple Bar Chart" ;  
  TITLE2 "#BYVAL(shift) Shift" ;  
  HBAR system /  
  DATALABEL  
  FILL  
  RESPONSE = usedgps  
  STAT    = sum  
;  
RUN;
```

Simple Bar Charts



Grouped Bar Charts



Grouped Bar Charts


SAS/GRAPH

```
PROC GCHART DATA = meanengs ;
BY date DESCENDING shift ;
TITLE "GCHART Grouped Bar Chart" ;
TITLE2 "#BYVAL(shift) Shift" ;
HBAR system / DISCRETE
SUMVAR=usedgps
SUM
SUMMLABEL = 'Engs'
GROUP = machine
PATTERNID = MIDPOINT
HTML = gpdrill
NOZERO
;
RUN;
*gpdrill=
CATS('HREF="gpplt',machnum,'.html"');
```


ODS Graphics

```
PROC SGPLOT DATA=meanengs
CYCLEATTRS ;
BY date DESCENDING shift ;
TITLE "SGPLOT Grouped Bar Chart" ;
TITLE2 "#BYVAL(shift) Shift" ;
HBAR system /
DATALABEL
FILL
GROUP = machine
RESPONSE = usedgps
STAT = sum
URL = urlgp
;
RUN;
* urlgp =
CATS('gpplt',machnum,'.html');
```

Grouped Bar Charts



Stacked Bar Charts



Stacked Bar Charts


SAS/GRAPH

```
PROC GCHART DATA=freq72go ;
BY date DESCENDING shift ;
  TITLE "GCHART Stacked Bar Chart" ;
  HBAR system / DISCRETE
    SUMVAR=gpwkld
    SUM
    SUMLABEL = "Engines"
    NOZERO
    SUBGROUP = wkldname
    PATTERNID = SUBGROUP
    HTML = gphtml
    HTML_LEGEND = bardrill
;
RUN;
```


ODS Graphics

```
PROC SGPLOT DATA=freq72go ;
BY date DESCENDING shift ;
  TITLE "SGPLOT Stacked Bar Chart" ;
  HBAR system / RESPONSE=gpwkld
    DATALABEL
    FILL
    GROUP = wkldname
    STAT = sum
    URL = urlwkchtgp
;
RUN;
```

Stacked Bar Charts



Combined Bar/Line Charts



Combined Bar/Line Charts

SAS/GRAPH

```
AXIS99 LABEL=NONE VALUE=NONE
  NOPLANE ;
PROC GBARLINE DATA=summwkld ;
  TITLE "GBARLINE Bar/Line Chart" ;
  BAR time / SUMVAR=scgps
    SUBGROUP   = wkldname
    ANNOTATE   = annowkld
    MAXIS      = axis99
    DISCRETE
    SPACE      = 0
;
  PLOT / SUMVAR=usedgps ;
RUN;
```

ODS Graphics

```
PROC SGPLOT DATA=summwkld
  SGANNO=annotime ;
  TITLE "SGPLOT Bar/Line Chart" ;
  VBAR time / RESPONSE=scgps
    GROUP=wkldname
    BARWIDTH=1 ;
  VLINE time / RESPONSE=usedgps
    GROUP=wkldname ;
  XAXIS DISPLAY=(NOLABEL NOTICKS
    NOVALUES) ;
RUN;
```

Time Axis Annotation


SAS/GRAPH

```
DATA annowkld ;
  SET summwkld (KEEP=date machine
    system time) ;
  IF MOD(time,7200) = 0 THEN DO;
    function = 'label' ;
    color    = 'black' ;
    position = '8'    ;
    when     = 'A'    ;
    STYLE    = 'simplex';
    hsys     = '5'    ;
    xsys     = '2'    ;
    ysys     = '1'    ;
    size     = 2.5    ;
    x        = time   ;
    y        = 0      ;
    text = PUT(time,TIME5.) ;
  END;
RUN;
```


ODS Graphics

```
DATA annotime ;
  function = 'TEXT' ;
  justify  = 'RIGHT' ;
  x1space  = 'DATAVALUE' ;
  y1space  = 'WALLPERCENT' ;
  y1       = -1.5    ;
  DO x1 = '00:00'T TO '22:00'T BY
    '02:00'T ;
    label = PUT(x1,TIME5.) ;
  OUTPUT;
END;
RUN;
```


Combined Bar/Line Charts



Scatter Plots



Scatter Plots


SAS/GRAPH

```
symbol1;  
PROC GPLOT DATA=summwkld ;  
  BY date machine system ;  
  TITLE "GPLOT Scatter Plot" ;  
  PLOT scgps*time=wkldname ;  
RUN;
```

ODS Graphics

```
PROC SGPLOT DATA=summwkld ;  
BY date machine system ;  
  TITLE "SGPLOT Scatter Plot" ;  
  SCATTER X=time Y=scgps /  
    GROUP=wkldname ;  
RUN;
```


Scatter Plots



Pie Charts

GCHART Pie Chart

FREQUENCY of Origin



Pie Charts

SAS/GRAPH


```
PROC GCHART DATA=sashelp.cars ;  
  TITLE 'GCHART Pie Chart' ;  
  PIE origin /  
    FILL=solid  
    SLICE=outside  
    VALUE=outside  
    COUTLINE=BLACK  
;  
RUN;
```

ODS Graphics


```
PROC TEMPLATE;  
  DEFINE STATGRAPH simplepie;  
    BEGINGRAPH;  
    ENTRYTITLE 'SGRENDER Pie Chart' ;  
    LAYOUT REGION;  
      PIECHART CATEGORY=ORIGIN /  
        DATALABELLOCATION=OUTSIDE;  
    ENDLAYOUT;  
  ENDGRAPH;  
END;  
RUN;
```

```
PROC SGRENDER DATA=sashelp.cars  
  TEMPLATE=simplepie;  
RUN;
```

Pie Charts



Stacked Plots



Stacked Plots


SAS/GRAPH

```
PROC GPLOT DATA=cpr70lp ;
  BY date machine system ;
  FORMAT sczips sczipa lparzips 3. ;
  TITLE "GPLOT Stacked Plot" ;
  PLOT (sczips sczipa lparzips)*time /
  OVERLAY
  AREAS = 2
  LEGEND
  HTML_LEGEND = combrpt
  HTML      = combrpt
;
RUN;
```

ODS Graphics

```
PROC SGPLOT DATA=cpr70lp ;
  TITLE "SGPLOT Stacked Plot" ;
  BY date machine system ;
  BAND X=time LOWER=0
  UPPER=sczips / FILL
  FILLATTRS=(COLOR=CX0000FF)
  LEGENDLABEL='zIIP Act' ;
  BAND X=time LOWER=sczips
  UPPER=sczipa / FILL
  FILLATTRS=(COLOR=CXFF66CC)
  LEGENDLABEL='zIIP Elig' ;
  SERIES X=time Y=lparzips /
  LINEATTRS=(PATTERN=1
  COLOR=CX0000FF)
  URL=urlcombrpt ;
  YAXIS MINOR LABEL='Engines'
  INTEGER ;
RUN;
```


Stacked Plots




Extras

Panels

GCHART Grouped Bar Chart
Prime Shift



GCHART Grouped Bar Chart
Night Shift



Panels


SAS/GRAPH

```
PROC GCHART DATA = meanengs ;
BY date DESCENDING shift ;
TITLE "GCHART Grouped Bar Chart" ;
TITLE2 "#BYVAL(shift) Shift" ;
HBAR system / DISCRETE
SUMVAR=usedgps
SUM
SUMLABEL = 'Engs'
GROUP = machine
PATTERNID = MIDPOINT
HTML = gpdrill
NOZERO ;
RUN;
```

ODS Graphics

```
ODS GRAPHICS ON / WIDTH=900px ;
PROC SGPanel DATA=meanengs
CYCLEATTRS ;
BY date ;
TITLE "SGPanel Grouped Chart" ;
PANELBY shift machine /
LAYOUT=lattice
UNISCALE=column
NOVARNAME ;
HBAR system / RESPONSE=usedgps
DATALABEL
URL=urlgp ;
COLAXIS OFFSETMIN=0;
RUN;
```

Panels



Create Custom Data Tips

```
PROC SGPLOT DATA= pimean ;  
  BY date ;  
  SCATTER X=srvcper Y=sysplexsys /  
    GROUP=mappi  
    URL=urlpi  
    MARKERATTRS=(SIZE=15px SYMBOL=circlefilled) ;  
  XAXIS DISCRETEORDER=unformatted ;  
RUN;
```

```
PROC TEMPLATE;  
  DEFINE STATGRAPH pisumplot ;  
  BEGINGRAPH;  
    ENTRYTITLE "SGRENDER PI SUMMARY for " _BYVAL_ ;  
    LAYOUT overlay ;  
    SCATTERPLOT X=srvcper Y=sysplexsys /  
      GROUP=mappi  
      URL=urlpi  
      NAME='pisumm'  
      MARKERATTRS=(SIZE=15px SYMBOL=circlefilled)  
      ROLENAMES=(  
        _pi=perindx _plex=sysplex  
        _sys=system _svcl=srvclass _per=period)  
      TIP=(  
        _plex _sys _svcl _per _pi)  
      DISCRETELEGEND 'pisumm' / TITLE="PI Level" ;  
    ENDLAYOUT;  
  ENDGRAPH;  
END;  
RUN;
```

```
PROC SGRENDER DATA=pimean TEMPLATE=pisumplot ;  
  BY date ;  
RUN;
```

Wrap Up

SAS ODS Graphics Manuals

- All SAS 9.3 manuals are available at <http://support.sas.com/documentation/online/doc/base/index.html>
- Under SAS Procedures Guides, see
 - SAS 9.3 Output Delivery System: User's Guide
 - SAS 9.3 ODS Graphics: Procedures Guide
- Under ODS Graphics, see
 - SAS 9.3 Graph Template Language: User's Guide
 - SAS 9.3 Graph Template Language: Reference

Recommended Reading

- Statistical Graphics in SAS, Warren F. Kuhfeld, 2010, ISBN 978-1-60764-485-9
- Using PROC SGPLOT for Quick High Quality Graphics, Delwiche and Slaughter,
<http://www.wuss.org/proceedings08/08WUSS%20Proceedings/papers/how/how05.pdf>
- Graphics Sample Output Gallery,
http://support.sas.com/sassamples/graphgallery/PROC_SGPLOT.html
- SAS Notes and Concepts for ODS,
<http://support.sas.com/rnd/base/ods/templateFAQ/index.html>